

NAG Library Routine Document

F08QVF (ZTRSYL)

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

F08QVF (ZTRSYL) solves the complex triangular Sylvester matrix equation.

2 Specification

```

SUBROUTINE F08QVF(TRANA, TRANB, ISGN, M, N, A, LDA, B, LDB, C, LDC,
1 SCAL, INFO)
INTEGER          ISGN, M, N, LDA, LDB, LDC, INFO
double precision SCAL
complex*16      A(LDA,*), B(LDB,*), C(LDC,*)
CHARACTER*1      TRANA, TRANB

```

The routine may be called by its LAPACK name ***ztrsyl***.

3 Description

F08QVF (ZTRSYL) solves the complex Sylvester matrix equation

$$\text{op}(A)X \pm X\text{op}(B) = \alpha C,$$

where $\text{op}(A) = A$ or A^H , and the matrices A and B are upper triangular; α is a scale factor (≤ 1) determined by the routine to avoid overflow in X ; A is m by m and B is n by n while the right-hand side matrix C and the solution matrix X are both m by n . The matrix X is obtained by a straightforward process of back-substitution (see Golub and Van Loan (1996)).

Note that the equation has a unique solution if and only if $\alpha_i \pm \beta_j \neq 0$, where $\{\alpha_i\}$ and $\{\beta_j\}$ are the eigenvalues of A and B respectively and the sign (+ or $-$) is the same as that used in the equation to be solved.

4 References

Golub G H and Van Loan C F (1996) *Matrix Computations* (3rd Edition) Johns Hopkins University Press, Baltimore

Higham N J (1992) Perturbation theory and backward error for $AX - XB = C$ *Numerical Analysis Report* University of Manchester

5 Parameters

1: TRANA – CHARACTER*1 *Input*

On entry: specifies the option $\text{op}(A)$.

TRANA = 'N'

$\text{op}(A) = A$.

TRANA = 'C'

$\text{op}(A) = A^H$.

Constraint: TRANA = 'N' or 'C'.

- 2: TRANB – CHARACTER*1 Input
On entry: specifies the option $\text{op}(B)$.
TRANB = 'N'
 $\text{op}(B) = B$.
TRANB = 'C'
 $\text{op}(B) = B^H$.
Constraint: TRANB = 'N' or 'C'.
- 3: ISGN – INTEGER Input
On entry: indicates the form of the Sylvester equation.
ISGN = +1
The equation is of the form $\text{op}(A)X + X \text{op}(B) = \alpha C$.
ISGN = -1
The equation is of the form $\text{op}(A)X - X \text{op}(B) = \alpha C$.
Constraint: ISGN = +1 or -1.
- 4: M – INTEGER Input
On entry: m , the order of the matrix A , and the number of rows in the matrices X and C .
Constraint: $M \geq 0$.
- 5: N – INTEGER Input
On entry: n , the order of the matrix B , and the number of columns in the matrices X and C .
Constraint: $N \geq 0$.
- 6: A(LDA,*) – **complex*16** array Input
Note: the second dimension of the array A must be at least $\max(1, M)$.
On entry: the m by m upper triangular matrix A .
- 7: LDA – INTEGER Input
On entry: the first dimension of the array A as declared in the (sub)program from which F08QVF (ZTRSYL) is called.
Constraint: $LDA \geq \max(1, M)$.
- 8: B(LDB,*) – **complex*16** array Input
Note: the second dimension of the array B must be at least $\max(1, N)$.
On entry: the n by n upper triangular matrix B .
- 9: LDB – INTEGER Input
On entry: the first dimension of the array B as declared in the (sub)program from which F08QVF (ZTRSYL) is called.
Constraint: $LDB \geq \max(1, N)$.
- 10: C(LDC,*) – **complex*16** array Input/Output
Note: the second dimension of the array C must be at least $\max(1, N)$.
On entry: the m by n right-hand side matrix C .
On exit: C is overwritten by the solution matrix X .

- 11: LDC – INTEGER Input
On entry: the first dimension of the array C as declared in the (sub)program from which F08QVF (ZTRSYL) is called.
Constraint: $LDC \geq \max(1, M)$.
- 12: SCAL – **double precision** Output
On exit: the value of the scale factor α .
- 13: INFO – INTEGER Output
On exit: INFO = 0 unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

INFO < 0

If INFO = $-i$, argument i had an illegal value. An explanatory message is output, and execution of the program is terminated.

INFO = 1

A and B have common or close eigenvalues, perturbed values of which were used to solve the equation.

7 Accuracy

Consider the equation $AX - XB = C$. (To apply the remarks to the equation $AX + XB = C$, simply replace B by $-B$.)

Let $\tilde{X}$ be the computed solution and R the residual matrix:

$$R = C - (A\tilde{X} - \tilde{X}B).$$

Then the residual is always small:

$$\|R\|_F = O(\epsilon)(\|A\|_F + \|B\|_F)\|\tilde{X}\|_F.$$

However, $\tilde{X}$ is **not** necessarily the exact solution of a slightly perturbed equation; in other words, the solution is not backwards stable.

For the forward error, the following bound holds:

$$\|\tilde{X} - X\|_F \leq \frac{\|R\|_F}{\text{sep}(A, B)}$$

but this may be a considerable over estimate. See Golub and Van Loan (1996) for a definition of $\text{sep}(A, B)$, and Higham (1992) for further details.

These remarks also apply to the solution of a general Sylvester equation, as described in Section 8.

8 Further Comments

The total number of real floating-point operations is approximately $4mn(m + n)$.

To solve the **general** complex Sylvester equation

$$AX \pm XB = C$$

where A and B are general matrices, A and B must first be reduced to Schur form (by calling F08PNF (ZGEES), for example):

$$A = Q_1 \tilde{A} Q_1^H \quad \text{and} \quad B = Q_2 \tilde{B} Q_2^H$$

where $\tilde{A}$ and $\tilde{B}$ are upper triangular and Q_1 and Q_2 are unitary. The original equation may then be transformed to:

$$\tilde{A}\tilde{X} \pm \tilde{X}\tilde{B} = \tilde{C}$$

where $\tilde{X} = Q_1^H X Q_2$ and $\tilde{C} = Q_1^H C Q_2$. $\tilde{C}$ may be computed by matrix multiplication; F08QVF (ZTRSYL) may be used to solve the transformed equation; and the solution to the original equation can be obtained as $X = Q_1 \tilde{X} Q_2^H$.

The real analogue of this routine is F08QHF (DTRSYL).

9 Example

This example solves the Sylvester equation $AX + XB = C$, where

$$A = \begin{pmatrix} -6.00 - 7.00i & 0.36 - 0.36i & -0.19 + 0.48i & 0.88 - 0.25i \\ 0.00 + 0.00i & -5.00 + 2.00i & -0.03 - 0.72i & -0.23 + 0.13i \\ 0.00 + 0.00i & 0.00 + 0.00i & 8.00 - 1.00i & 0.94 + 0.53i \\ 0.00 + 0.00i & 0.00 + 0.00i & 0.00 + 0.00i & 3.00 - 4.00i \end{pmatrix},$$

$$B = \begin{pmatrix} 0.50 - 0.20i & -0.29 - 0.16i & -0.37 + 0.84i & -0.55 + 0.73i \\ 0.00 + 0.00i & -0.40 + 0.90i & 0.06 + 0.22i & -0.43 + 0.17i \\ 0.00 + 0.00i & 0.00 + 0.00i & -0.90 - 0.10i & -0.89 - 0.42i \\ 0.00 + 0.00i & 0.00 + 0.00i & 0.00 + 0.00i & 0.30 - 0.70i \end{pmatrix}$$

and

$$C = \begin{pmatrix} 0.63 + 0.35i & 0.45 - 0.56i & 0.08 - 0.14i & -0.17 - 0.23i \\ -0.17 + 0.09i & -0.07 - 0.31i & 0.27 - 0.54i & 0.35 + 1.21i \\ -0.93 - 0.44i & -0.33 - 0.35i & 0.41 - 0.03i & 0.57 + 0.84i \\ 0.54 + 0.25i & -0.62 - 0.05i & -0.52 - 0.13i & 0.11 - 0.08i \end{pmatrix}.$$

9.1 Program Text

```
*      F08QVF Example Program Text
*      Mark 21 Release. NAG Copyright 2004.
*      .. Parameters ..
      INTEGER          NIN, NOUT
      PARAMETER        (NIN=5,NOUT=6)
      INTEGER          MMAX, NMAX
      PARAMETER        (MMAX=8,NMAX=8)
      INTEGER          LDA, LDB, LDC
      PARAMETER        (LDA=MMAX,LDB=NMAX,LDC=MMAX)
*      .. Local Scalars ..
      DOUBLE PRECISION SCAL
      INTEGER          I, IFAIL, INFO, J, M, N
*      .. Local Arrays ..
      COMPLEX *16      A(LDA,MMAX), B(LDB,NMAX), C(LDC,NMAX)
      CHARACTER        CLABS(1), RLABS(1)
*      .. External Subroutines ..
      EXTERNAL         X04DBF, ZTRSYL
*      .. Executable Statements ..
      WRITE (NOUT,*) 'F08QVF Example Program Results'
      WRITE (NOUT,*)
*      Skip heading in data file
      READ (NIN,*)
      READ (NIN,*) M, N
      IF (M.LE.MMAX .AND. N.LE.NMAX) THEN
*
*          Read A, B and C from data file
*
          READ (NIN,*) ((A(I,J),J=1,M),I=1,M)
          READ (NIN,*) ((B(I,J),J=1,N),I=1,N)
          READ (NIN,*) ((C(I,J),J=1,N),I=1,M)
*
*          Solve the Sylvester equation A*X + X*B = C for X
```

```

*
      CALL ZTRSYL('No transpose','No transpose',1,M,N,A,LDA,B,LDB,C,
+             LDC,SCAL,INFO)
      IF (INFO.GE.1) THEN
        WRITE (NOUT,99999)
        WRITE (NOUT,*)
      END IF
*
*      Print the solution matrix X
*
      IFAIL = 0
      CALL X04DBF('General',' ',M,N,C,LDC,'Bracketed','F7.4',
+             'Solution matrix X','Integer',RLABS,'Integer',
+             CLABS,80,0,IFAIL)
*
      WRITE (NOUT,*)
      WRITE (NOUT,99998) 'SCAL = ', SCAL
      ELSE
        WRITE (NOUT,*) 'MMAX and/or NMAX too small'
      END IF
*
99999 FORMAT ('/' A and B have common or very close eigenvalues.'/' Pe',
+           'rturbed values were used to solve the equations')
99998 FORMAT (1X,A,1P,E10.2)
      END

```

9.2 Program Data

F08QVF Example Program Data

```

4 4
(-6.00,-7.00) ( 0.36,-0.36) (-0.19, 0.48) ( 0.88,-0.25) :Values of M and N
( 0.00, 0.00) (-5.00, 2.00) (-0.03,-0.72) (-0.23, 0.13)
( 0.00, 0.00) ( 0.00, 0.00) ( 8.00,-1.00) ( 0.94, 0.53)
( 0.00, 0.00) ( 0.00, 0.00) ( 0.00, 0.00) ( 3.00,-4.00) :End of matrix A
( 0.50,-0.20) (-0.29,-0.16) (-0.37, 0.84) (-0.55, 0.73)
( 0.00, 0.00) (-0.40, 0.90) ( 0.06, 0.22) (-0.43, 0.17)
( 0.00, 0.00) ( 0.00, 0.00) (-0.90,-0.10) (-0.89,-0.42)
( 0.00, 0.00) ( 0.00, 0.00) ( 0.00, 0.00) ( 0.30,-0.70) :End of matrix B
( 0.63, 0.35) ( 0.45,-0.56) ( 0.08,-0.14) (-0.17,-0.23)
(-0.17, 0.09) (-0.07,-0.31) ( 0.27,-0.54) ( 0.35, 1.21)
(-0.93,-0.44) (-0.33,-0.35) ( 0.41,-0.03) ( 0.57, 0.84)
( 0.54, 0.25) (-0.62,-0.05) (-0.52,-0.13) ( 0.11,-0.08) :End of matrix C

```

9.3 Program Results

F08QVF Example Program Results

Solution matrix X

```

1 2 3 4
1 (-0.0611, 0.0249) (-0.0031, 0.0798) (-0.0062, 0.0165) ( 0.0054,-0.0063)
2 ( 0.0215,-0.0003) (-0.0155, 0.0570) (-0.0665, 0.0718) ( 0.0290,-0.2636)
3 (-0.0949,-0.0785) (-0.0415,-0.0298) ( 0.0357, 0.0244) ( 0.0284, 0.1108)
4 ( 0.0281, 0.1052) (-0.0970,-0.1214) (-0.0271,-0.0940) ( 0.0402, 0.0048)

```

SCAL = 1.00E+00